

Autostar Tour Programming

This document describes how to write your own custom tours for the Autostar handbox. These tours are in addition to the existing tours within the Autostar program.

What is a Custom Tour?

A custom Autostar tour consists of a list of specific objects, objects chosen by and presented in the order and style defined by the user. A tour can be many things. It can be used to introduce people to the wonders of the night sky or to enhance an astronomy lesson plan. It can be designed around a specific theme (i.e. brightest objects in the sky; compare open clusters; observe differences between spiral and elliptical galaxies, etc.) Custom tours are limited only by the imagination.

When a tour is selected, the telescope presents the objects in the order they are listed within the tour. If available, a description of each object is displayed. To view an object, the user presses GOTO and the telescope moves to that object. The user can move forward or backward in a tour using regular Autostar menu selection techniques. Before presenting an object in a tour, Autostar verifies that the object is far enough above the horizon to be visible. If an object is not high enough above the horizon, it automatically skips to the next visible object in the tour. The user may exit a tour at any time by pressing the MODE key.

Tour Objects

Tours can be created using objects from within the Autostar database or a user can enter in different objects (such as a newly discovered comet). Objects within a tour can be presented in two different modes: automatic mode or interactive mode. The author of the tour determines how the objects are presented by choosing one of these two modes when writing the tour.

Automatic Mode: As the user advances to each object, the title of the object appears on line 1 of the display while its descriptive text automatically starts scrolling across line 2.

Interactive Mode: As the user advances to each object, the tour name appears on line 1, while the name of the next object in the tour is displayed on line 2. To access an object's description in this mode, the user must press ENTER.

Tour Terminology

Writing a custom tour is similar to writing a script for a play – only this play will be performed by Autostar-equipped telescopes. Therefore it is necessary to learn the proper "stage directions", or how to direct the Autostar to do what is desired. A tour is basically an ASCII text file that contains a list of directions and descriptions. Tours can be written using any text editor or word processor, as long as the tour can be saved as a "text only" or "MS Dos text" file.

Each line of a tour will be either a comment line, a command line, or the continuation of a description string.

Comment lines: A comment line allows the writer to include information in his/her tour that they want to keep in the file, but don't want the user to see during the tour. Comment lines can include authorship credits, revision histories, copyrights, course and lesson numbers, etc. All comment lines begin with a '/' character in column 1 of the line. The example below consists of six comment lines taken from the script example contained in Appendix A of this document. (This example uses two "///", only one is necessary.)

```
////////////////////////////////////  
// Sample Tour Taken from September 99 Sky & Telescope  
//  
// (c) 1999 Meade Inst.
```

```
// (c) 1999 Sky Publishing  
////////////////////////////////////
```

Command lines: A command line can include the following: a keyword, RA and DEC coordinates, a title string and a description string.

Keyword: A keyword describes an action to be performed during the tour. Autostar recognizes the following keywords:

```
TITLE  
TEXT  
USER  
NGC  
IC  
SAO  
MESSIER  
CALDWELL  
PLANET  
MOON  
SATELLITE  
ASTEROID  
COMET  
LUNAR ECLIPSE  
METEOR SHOWER  
DEEP SKY  
CONSTELLATION  
STAR  
LANDMARK  
PICK ONE / PICK END  
AUTO SLEW ON / AUTO SLEW OFF  
#END
```

Keywords may be preceded by the qualifier "AUTO SELECT". If this qualifier appears on a command line, Autostar will execute this command automatically (in the Automatic Mode). If the qualifier is not present, the command will be executed in the Interactive Mode.

RA: Enter the right ascension of the desired object in the following format: HH:MM:SS.
For example: 18:51:05

DEC: Enter the declination of the desired object in the following format: DDdMMmSSs.
For example: -06d16m00s

Title string: Text within a title string is displayed as the title of the object. In Interactive mode, it appears on line 2 until it is selected with the ENTER key. In Automatic mode, or after interactive mode selection, the title string appears on line 1 as the description string scrolls across line 2. A title string must be surrounded by quotation marks. This string can contain up to 16 characters. For example: "M17" or "My Favorite Star".

Description string: Surrounded by quotation marks, text within this string appears as the description for an object. Because object descriptions may be longer than one line, it is necessary to end each line with a quotation mark and a hard return. Begin the next description line with a quotation mark. If you need a quotation mark within the description string, use two

quotes together "" at the beginning and end of the desired phrase. For example: "This nebula is considered ""awesome"" by many who view it."

Writing a Tour

From the list of terms above, it is now possible to write your own Autostar custom tour. Following is a list of command lines, complete with keywords and necessary strings.

Placing the word AUTO SELECT before any of the following command lines activates the automatic mode and when selected, Autostar then automatically searches and finds the designated object.

Command Line Examples:

TITLE

Title must be the first keyword in your tour after any comment lines. The title of your tour should be 15 characters or less. This title will be displayed by Autostar when the system is asked what tours are available. For example: TITLE "A Star's Life"

TEXT "title string" "description string"

This command line allows you to display a text title and description.

USER ra dec "title string" "description string"

This command line allows you to access a specific object with your own description. Enter USER, then the RA and DEC of a desired object, and its title and description. Use the format described in the Tour Terminology section above.

The commands below specify objects that are already in the Autostar handbox. If these commands follow the command AUTO SELECT, the object's title will appear on line 1 while its description scrolls across line 2.

Do not add a description string after the following command lines. These commands access objects with existing descriptions in the Autostar database.

NGC xxxx

Enter NGC followed by the desired NGC number and Autostar provides the user with a description of the object from its database. For example: NGC 4256

IC xxxx

Enter IC followed by the desired Index Catalogue number and Autostar provides the user with a description of the object from its database. For example: IC 1217

SAO xxxxxx

Enter SAO followed by the desired SAO number and Autostar provides the user with the description of the object from its database. For example: SAO 651702

MESSIER xxx

Enter MESSIER followed by the desired Messier number and Autostar provides the user with the description of the object from its database. For example: MESSIER 42

CALDWELL xxx

Enter CALDWELL followed by the desired Caldwell number and Autostar provides the user with the description of the object from its database. For example: CALDWELL 17

PLANET "name"

Enter PLANET and then the name of the desired planet in quotes. Autostar then provides the user with the description of the selected planet from its database. For example: PLANET "Jupiter"

MOON

The MOON command accesses the Moon's information from Autostar's database.

SATELLITE "name"

Enter SATELLITE and then the name of the desired satellite in quotes. Autostar then displays information from its satellite database.

ASTEROID "name"

Enter ASTEROID and then the name of the desired asteroid in quotes. Autostar then displays information from its asteroid database. For example: ASTEROID "Ceres"

COMET "name"

Enter COMET and then the name of the desired comet in quotes. Autostar then displays information from its comet database. For example: COMET "Halley"

LUNAR ECLIPSE

If LUNAR ECLIPSE is a part of the tour, Autostar checks its database every time the tour is activated to see if there is a lunar eclipse visible that evening. If no eclipse is visible, the Lunar Eclipse option is skipped and the tour proceeds to the next option.

METEOR SHOWER

If METEOR SHOWER is a part of the tour, Autostar checks its database every time the tour is activated to see if there is a meteor shower that evening. If no meteor shower is visible, the option is skipped and the tour proceeds to the next option.

DEEP SKY "name"

Enter DEEP SKY followed by the name of the desired object in quotes. For example: DEEP SKY "Small Magellanic Cloud"

CONSTELLATION "name"

Enter CONSTELLATION followed by the name of the desired constellation in quotes. For example: CONSTELLATION "Orion"

STAR "name"

Enter STAR followed by the name of the desired star in quotes. For example: STAR "Rigel"

LANDMARK az alt "title" "description"

Enter the azimuth (az) for the desired object in the following format xxxdxxmxxs. For example: 123d27m00s. Then, enter the altitude for the desired object in the following format: xdxmxxs. Then, enter the title string and description string in quotes. For example: LANDMARK 123d27m00s 57d20m20s "Landmark 1" "North corner of" "apartment building"

PICK ONE / PICK END

These two statements are used to surround a list of items that Autostar can choose from during a tour. Autostar begins at the top of the PICK ONE list and displays the first object from the list that is above the horizon and ignores the rest.

This statement is useful for developing tours that can be presented year round. For each object type you wish to use to illustrate your tour, include 10 to 12 examples, spaced across the range of right ascension. Bracket them by the PICK ONE / PICK END statements and exactly one current example will be shown to the user. For example:

```
AUTO SELECT TEXT "Globular Cluster" "Globular clusters are huge balls of stars."
"They can contain 50,000 to 100,000 stars and are located on the fringes of our"
"galaxy."
PICK ONE
AUTO SELECT MESSIER 13
AUTO SELECT MESSIER 15
AUTO SELECT MESSIER 92
AUTO SELECT MESSIER 4
AUTO SELECT MESSIER 68
AUTO SELECT NGC 2419
AUTO SELECT TEXT "None Available" "I'm sorry. There are no bright globular"
"clusters visible at this time."
PICK END
```

AUTO SLEW ON / AUTO SLEW OFF

With AUTO SLEW ON in the tour, Autostar automatically slews to the objects first, before displaying the text description. This feature is helpful when designing tours in which observing certain objects is required. For example: an astronomy professor may require his students to observe the first four objects on a tour, then have the last two objects for extra credit. He would place AUTO SLEW ON before his first required object and AUTO SLEW OFF after his fourth object. The students would have to manually slew to the last two extra credit objects.

#END

To end a tour, type the command #END on a separate line at the very end of the tour.

Downloading Tours

Once a tour is written and stored as an ASCII text file (saved as either a "text only" or "MS Dos text" file), it can be loaded into the Autostar using the Autostar Update Utility on your PC. As tours are downloaded into the handbox, Autostar examines the programming. If it doesn't understand any of the terminology used within a tour, it flags these questionable areas and displays them in a pop-up window on your computer screen. From there, you can make the necessary corrections and try the download again.